mtaylor.mw "multitaylor", usean muuttujan Taylorin polynomi

7.3.2002 HA Kurssilla InfoV2

Palauttaminen 1. muuttujan tapaukseen.

```
> restart:
> with(plots):
> setoptions3d(axes=boxed,orientation=[-30,50]):
```

Taylorin lause

Kahden muuttujan Taylorin lauseen johto Maplella:
Voimme toistaa Maplella sen, mita teimme eilen taululla. Homma toimii mainiosti, koska Maple osaa ketjusaannon.
Olkoon (a,b) tason piste ja (h1,h2) lisaysvektori. Nama ovat koko tarkastelun ajan kiinteita.

```
> a:='a':b:='b':h1:='h1':h2:='h2':f:='f':
> F:=t->f(a+t*h,b+t*k);
```

$$F := t \to f(a + t\,h, b + t\,k) \tag{1.1}$$

```
> F(0),D(F)(0),expand((D@@2)(F)(0));
```

$$f(a, b), \mathrm{D}_1(f)(a, b)\,h + \mathrm{D}_2(f)(a, b)\,k, \mathrm{D}_{1,1}(f)(a, b)\,h^2 + 2\,h\,\mathrm{D}_{1,2}(f)(a, b)\,k + \mathrm{D}_{2,2}(f)(a, b)\,k^2 \tag{1.2}$$

```
> T4:=add((1/j!*expand((D@@j)(F)(0)),j=0..4));
```

$$T4 := f(a, b) + \mathrm{D}_1(f)(a, b)\,h + \mathrm{D}_2(f)(a, b)\,k + \frac{1}{2}\,\mathrm{D}_{1,1}(f)(a, b)\,h^2 + h\,\mathrm{D}_{1,2}(f)(a, b)\,k + \frac{1}{2}\,\mathrm{D}_{2,2}(f)(a, b)\,k^2 + \frac{1}{6}\,\mathrm{D}_{1,1,1}(f)(a, b)\,h^3 + \frac{1}{2}\,h^2\,\mathrm{D}_{1,1,2}(f)(a, b)\,k + \frac{1}{2}\,h\,\mathrm{D}_{1,2,2}(f)(a, b)\,k^2 + \frac{1}{6}\,\mathrm{D}_{2,2,2}(f)(a, b)\,k^3 + \frac{1}{24}\,\mathrm{D}_{1,1,1,1}(f)(a, b)\,h^4 + \frac{1}{6}\,h^3\,\mathrm{D}_{1,1,1,2}(f)(a, b)\,k + \frac{1}{4}\,h^2\,\mathrm{D}_{1,1,2,2}(f)(a, b)\,k^2 + \frac{1}{6}\,h\,\mathrm{D}_{1,2,2,2}(f)(a, b)\,k^3 + \frac{1}{24}\,\mathrm{D}_{2,2,2,2}(f)(a, b)\,k^4 \tag{1.3}$$

Esim: Approksimoi funktiota

```
> f:=(x,y)->sqrt(x^2+y^3);
```

$$f := (x, y) \to \sqrt{x^2 + y^3} \tag{1.4}$$

2. asteen Taylorin polynomilla. Arvioi silla lukua.

```
> 'sqrt('1.02'^2+'1.97'^3)';
```

(1.5)

$$\sqrt{\text{'1.02'}^2 + \text{'1.97'}^3} \qquad (1.5)$$

Tapa 1: Puhdas "kasinlasku"

```
> f:=(x,y)->sqrt(x^2+y^3); p:=1,2:
```

$$f := (x, y) \rightarrow \sqrt{x^2 + y^3} \qquad (1.6)$$

```
> D1f:=D[1](f)(p);D2f:=D[2](f)(p);
```

$$D1f := \frac{1}{3}$$

$$D2f := 2 \qquad (1.7)$$

```
> D11f:=D[1,1](f)(p);D12f:=D[1,2](f)(p);D22f:=D[2,2](f)(p);
```

$$D11f := \frac{8}{27}$$

$$D12f := -\frac{2}{9}$$

$$D22f := \frac{2}{3} \qquad (1.8)$$

```
> T2:=f(p)+D1f*h1+D2f*h2+1/2*D11f*h1^2+D12f*h1*h2+1/2*D22f*h2^2;
```

$$T2 := 3 + \frac{1}{3}\,h1 + 2\,h2 + \frac{4}{27}\,h1^2 - \frac{2}{9}\,h1\,h2 + \frac{1}{3}\,h2^2 \qquad (1.9)$$

Sijoitetaan nyt lukuarvot T2:een:

```
> subs(h1=.02,h2=-.03,T2);
```

$$2.947159259 \qquad (1.10)$$

```
> tarkka:=f(1.02,1.97);
```

$$tarkka := 2.947163552 \qquad (1.11)$$

```
> %-%%;
```

$$0.000004293 \qquad (1.12)$$

Varsin vaikuttava tarkkuus.
Usein haluamme Taylorin polynomin x:n ja y:n avulla:

```
> T2xy:=subs(h1=x-1,h2=y-2,T2);
```

$$T2xy := -\frac{4}{3} + \frac{1}{3}\,x + 2\,y + \frac{4}{27}\,(x-1)^2 - \frac{2}{9}\,(x-1)\,(y-2) + \frac{1}{3}\,(y-2)^2 \qquad (1.13)$$

Maplen automaattinen sieventaja on hiema liian innokas, kertoo vakisin 1. asteen termit auki.

Kaytetaan sitten valmista **mtaylor**-funktiota:

```
> f(x,y);
```

$$\sqrt{x^2 + y^3} \qquad (1.14)$$

```
> mtaylor(f(x,y),[x=1,y=2],3);
```

$$-\frac{4}{3} + \frac{1}{3}\,x + 2\,y + \frac{4}{27}\,(x-1)^2 - \frac{2}{9}\,(x-1)\,(y-2) + \frac{1}{3}\,(y-2)^2 \qquad (1.15)$$

Sama harmi 1. asteen termeihin nahden. "Kauneusvirhe" voidaan hoitaa palaamalla h1h2-muotoon:

```
> subs(x=1+h1,y=2+h2,%);
```

$$3 + \frac{1}{3}\, h1 + 2\, h2 + \frac{4}{27}\, h1^2 - \frac{2}{9}\, h1\, h2 + \frac{1}{3}\, h2^2 \qquad \textbf{(1.16)}$$

>

Piirretaan funktiopinta ja Taylor-approksimaatio

> $H := 1 : display(plot3d(f(x, y), x = 1 - H..1 + H, y = 2 - H..2 + H, color = blue),$
> $plot3d(T2xy, x = 1 - H..1 + H, y = 2 - H..2 + H, color = red));$



>

Vaikka H=1 (niinkin suuri), niin kuvan reunojen lahella juuri eroavat. Pyorita kuvaa hiirella! (Maple-tyoarkilla onnistuu, pdf:ssa ei.)

Usein kannattaa piirtaa erotuksen kuvaaja:

> $plot3d(f(x, y) - T2xy, x = 1 - H..1 + H, y = 2 - H..2 + H);$


0.15
0.10
0.05
0
−0.05
−0.10
−0.15
0
0.5
x
1
1.5
2
1.5
2
y
2.5
3

>